Práctica 5. "Funciones de Dos Variables"

Dibujado (Ejercicio 1)

El comando: Plot3D[f,{x,a,b},{y,c,d}] permite representar funciones de dos variables definidas sobre un dominio rectangular *[a,b] x[c,d]*. Por ejemplo:

▪ Apartado (a)

```
Plot3D[x^2 + y^2, {x, -2, 2}, {y, -3, 3}]
```



▪ Apartado (b)

En este caso el dominio es el exterior de una circunferencia de centro el origen y radio 2:

```
Plot3D[Log[x^2 + y^2 - 4], {x, -5, 5}, {y, -5, 5}]
```



Mediante la opcion **PlotPoints** (cuyo valor por defecto es 15) es posible modificar la resolución de la figura obtenida. Por ejemplo:

```
Plot3D[Log[x^2 + y^2 - 4], {x, -5, 5}, {y, -5, 5}, PlotPoints -> 60]
```



▪ Apartado (d)

```
Plot3D[-Sqrt[(Log[x + y]) ^ 2 + 1], {x, -2, 2}, {y, -2, 2}]
```



Tambien se puede modificar el punto de vista de la figura mediante la opción **ViewPoint**. Esta opción se encuentra en **3DViewPointSelector**.

```
Plot3D[-Sqrt[(Log[x + y]) ^ 2 + 1], {x, -2, 2}, {y, -2, 2},
 ViewPoint -> {1.501, 1.483, 2.645}]
```



▪ Apartado (c)

En este caso el dominio está formado por intervalos en los cuadrantes primero y tercero (con -5< =x<=0, y 0>=x<=5 respectivamente):

```
Plot3D[ArcSin[x / 5] + Sqrt[x y], {x, -6, 6}, {y, -6, 6}]
```



Aumentando el número de puntos de dibujado (con la opción **PlotPoints**) se mejora la calidad del dibujo:

```
Plot3D[ArcSin[x / 5] + Sqrt[x y], {x, -6, 6}, {y, -6, 6}, PlotPoints -> 80]
```




```
a = 10;
Plot3D[x ^ 2 / 4 + y ^ 2, {x, -a, a}, {y, -a, a}]
```



Límites dobles (Ejercicio 10)

■ Apartado (b)

Mathematica no incorpora comandos para el cálculo de límites dobles. La única posibilidad consiste en aplicar límites reiterados o cualesquiera otros que involucren sólo funciones de una variable. Aquí analizamos los límites reiterados.

Por ejemplo:

```
f[x_] := (x ^ 2 - y ^ 2) / (x ^ 2 + y ^ 2)
Limit[Limit[f[x], x -> 0], y -> 0]
-1
Limit[Limit[f[x], y -> 0], x -> 0]
1
```

Salen distintos. Luego el límite doble no existe.

Limites por rectas (ejercicio 7)

Si se desea calcular límites por rectas y=m x+n, se aplica una sustitución y->m x+n. Por ejemplo:

■ Apartado (a)

Se calcula el límite por rectas pasando por el origen (de la forma y=a x):

```
Limit[y / (y + x^2) /. y -> a x, x -> 0]
```

1

Límite por parábolas de la forma y=a x^2:

```
Limit[y / (y + x^2) /. y -> a x^2, x -> 0]
```

$\frac{a}{1+a}$

- **Apartado (b)**

Se calcula el límite por rectas pasando por el origen (de la forma y=a x):

```
Limit[x y^2 / (y^4 + x^2) /. y -> a x, x -> 0]
```

0

Límite por funciones de la forma y=a $\sqrt{x}$:

```
Limit[x y^2 / (y^4 + x^2) /. y -> a Sqrt[x], x -> 0]
```

$\frac{a^2}{1+a^4}$

Limites por rectas (ejercicio 8)

- **Apartado (a)**

Se calcula el límite por rectas pasando por el origen (de la forma y=a x):

```
Limit[3 x y^2 / (y^4 + 2 x^2) /. y -> a x, x -> 0]
```

0

Límite doble hecho por el cambio a polares:

```
Limit[3 x y^2 / (y^4 + 2 x^2) /. {x -> r Cos[t], y -> r Sin[t]}, r -> 0]
```

0

- **Apartado (b)**

Se calcula el límite por rectas pasando por el origen (de la forma y=a x):

```
Limit[x^2 / Sqrt[x^2 + y^2] /. y -> a x, x -> 0]
```

0

Límite doble hecho por el cambio a polares:

```
Limit[x^2 / Sqrt[x^2 + y^2] /. {x -> r Cos[t], y -> r Sin[t]}, r -> 0]
```

0

- **Apartado (c)**

Se calcula el límite por rectas pasando por el origen (de la forma y=a x):

```
Limit[x^2 y / (x^4 + y^2) /. y -> a x, x -> 0]
```

0

Límite doble hecho por el cambio a polares:

```
Limit[x^2 y / (x^4 + y^2) /. {x -> r Cos[t], y -> r Sin[t]}, r -> 0]
```

0

Limites por polares (ejercicio 11)

Apartado (a)

Límite doble hecho por el cambio a polares:

```
Limit[(x^3 + y^3) / (2 y^3) /. {x -> r Cos[t], y -> r Sin[t]}, r -> 0]
```

$\frac{1}{2}\,\mathrm{Csc}[t]^3\,(\mathrm{Cos}[t]^3 + \mathrm{Sin}[t]^3)$

El límite depende del ángulo t. Por tanto, no existe el límite doble.

Apartado (b)

```
Limit[(x^4 + 2 y^2) / (x^2 + y^2) /. {x -> r Cos[t], y -> r Sin[t]},
 r -> 0]
```

$\frac{2\,\mathrm{Sin}[t]^2}{\mathrm{Cos}[t]^2 + \mathrm{Sin}[t]^2}$

Por tanto, no existe el límite doble.

Apartado (c)

```
Limit[(x^3 + y^3) / (x^2 + y^2) /. {x -> r Cos[t], y -> r Sin[t]},
 r -> 0]
```

0

Apartado (d)

```
Limit[x^2 / Sqrt[x^2 + y^2] + y^2 /. {x -> r Cos[t], y -> r Sin[t]},
 r -> 0]
```

0

Derivadas parciales

El comando D[f,x] permite calcular la derivada parcial de la funcion f respecto a la variable indicada. Por ejemplo:

```
D[Sin[x y], x]
```

y Cos[x y]

Definimos una función de dos variables:

```
f[x_, y_] := Exp[x] y^2
```

Sus derivadas parciales se calculan como:

```
{D[f[x, y], x], D[f[x, y], y]}
```

$\{E^x y^2,\ 2 E^x y\}$

■ Vector gradiente

Definimos el vector gradiente en la forma:

```
Gradiente[f_, v_List] := D[f, #] & /@ v
```

Por ejemplo:

```
Gradiente[f[x, y], {x, y}]
```

$\{E^x y^2, 2 E^x y\}$

Esta definición del vector gradiente es independiente del nombre de la función y de sus variables:

```
Gradiente[u^3 Sin[u Exp[v]], {u, v}]
```

$\{E^v u^3 \text{Cos}[E^v u] + 3 u^2 \text{Sin}[E^v u], E^v u^4 \text{Cos}[E^v u]\}$

Puntos criticos

Los puntos criticos son:
-puntos donde no hay derivadas parciales (puntos *singulares*)
- puntos donde las derivadas primeras se hacen cero (puntos *regulares*)
- puntos *frontera*.

Si suponemos que las funciones con las que trabajamos son siempre diferenciables, **nunca tenemos puntos singulares**.

Si se consideran extremos **libres**, sólo buscamos puntos regulares, es decir, que anulan al gradiente. El siguiente comando permite calcular los puntos críticos de una función en las variables indicadas:

```
PuntosCriticos[f_, v_List] :=
    Module[{n = Length[v], l},
        l = Table[0, {i, 1, n}];
    Solve[Gradiente[f, v] == l]
    ]
```

Por ejemplo:

```
l = PuntosCriticos[x^2 y^3 + 2 x - 3 y, {x, y}]
```

$\{\{x \to -1, y \to 1\}, \{x \to -I, y \to I\}, \{x \to I, y \to -I\}, \{x \to 1, y \to -1\}\}$

A continuación, es necesario determinar si los puntos críticos obtenidos son o no extremos. En el caso de extremos libres, esto se consigue con el comando:

```
MatrizHessiana[f_, {x_, y_}] :=
Module[{},
{{D[f, {x, 2}], D[f, {x, 1}, {y, 1}]},
        {D[f, {x, 1}, {y, 1}], D[f, {y, 2}]}}
    ]
```

que calcula la matrix Hessiana de la función indicada.

```
MatrizHessiana[x^2 y^3 + 2 x - 3 y, {x, y}]
```

$\{\{2 y^3, 6 x y^2\}, \{6 x y^2, 6 x^2 y\}\}$

A continuación, se evalua dicha matriz en los puntos críticos:

```
% /. l
{{{2, -6}, {-6, 6}}, {{-2 I, 6 I}, {6 I, -6 I}},
 {{2 I, -6 I}, {-6 I, 6 I}}, {{-2, 6}, {6, -6}}}
```

El determinante de estas matrices resulta:

```
Det[#] & /@ %
{-24, 24, 24, -24}
```

Por tanto, los puntos críticos asociados a valores reales (los complejos no se tienen en cuenta) obtenidos corresponden a puntos de silla.

```
Plot3D[x^2 y^3 + 2 x - 3 y, {x, -2, 2}, {y, -2, 2}]
```




```
- SurfaceGraphics -
ContourPlot[x^2 y^3 + 2 x - 3 y, {x, -2, 2}, {y, -2, 2}]
```



Ejercicio 36

En este ejercicio se obtienen 3 puntos críticos reales:

```
pc36 = PuntosCriticos[2 x^2 + y^2 + x^2 y, {x, y}]
{{y → -2, x → -2}, {y → -2, x → 2}, {y → 0, x → 0}}
```

La matriz Hessiana resulta:

```
MatrizHessiana[2 x^2 + y^2 + x^2 y, {x, y}]
{{4 + 2 y, 2 x}, {2 x, 2}}
```

Evaluada en los puntos, se obtiene:

```
% /. pc36
{{{0, -4}, {-4, 2}}, {{0, 4}, {4, 2}}, {{4, 0}, {0, 2}}}
```

con determinante:

```
Det[#] & /@ %
{-16, -16, 8}
```

Por tanto, los puntos (-2,-2) y (-2,2) son de silla. En cambio, el punto (0,0) es un punto de extremo. Para saber de que tipo, basta ver su matriz hessiana:

```
MatrizHessiana[2 x^2 + y^2 + x^2 y, {x, y}] /. {x -> 0, y -> 0}
{{4, 0}, {0, 2}}
```

Como A=4>0, es un punto de mínimo.

```
Plot3D[2 x^2 + y^2 + x^2 y, {x, -3, 3}, {y, -3, 3}]
```



Ejercicio 40

Analizamos ahora el caso de extremos condicionados.

En este caso, *f(x,y)=x y* es la función problema:

```
Clear[f];
f[x_, y_] := x y
```

la restriccion viene dada por:*p(x,y)=x^2+y^2-1*

```
p[x_, y_] := x^2 + y^2 - 1
```

La función de Lagrange viene dada por:

```
F[x_, y_, p_] := f[x, y] + m p[x, y]
```

donde p es el multiplicador.

Obtenemos los puntos críticos de la función de Lagrange, dada por:

```
pc40 = PuntosCriticos[F[x, y, p], {x, y, m}]
```

$$\left\{\left\{m \to -\frac{1}{2},\ x \to -\frac{1}{\sqrt{2}},\ y \to -\frac{1}{\sqrt{2}}\right\},\ \left\{m \to -\frac{1}{2},\ x \to \frac{1}{\sqrt{2}},\ y \to \frac{1}{\sqrt{2}}\right\},\ \left\{m \to \frac{1}{2},\ x \to -\frac{1}{\sqrt{2}},\ y \to \frac{1}{\sqrt{2}}\right\},\ \left\{m \to \frac{1}{2},\ x \to \frac{1}{\sqrt{2}},\ y \to -\frac{1}{\sqrt{2}}\right\}\right\}$$

Los puntos tridimensionales resultan:

```
ptos3D = {x, y, f[x, y]} /. %
```

$\left\{\left\{-\frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}, \frac{1}{2}\right\}, \left\{\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, \frac{1}{2}\right\}, \left\{-\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, -\frac{1}{2}\right\}, \left\{\frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}, -\frac{1}{2}\right\}\right\}$

Como el recinto es compacto (cerrado y acotado), los máximos y mínimos absolutos se obtienen evaluando la función en dichos puntos.

```
f[x, y] /. pc40
```

$\left\{\frac{1}{2}, \frac{1}{2}, -\frac{1}{2}, -\frac{1}{2}\right\}$

Por tanto, los dos primeros puntos corresponden a máximos absolutos y los dos últimos a mínimos absolutos.

A continuación, dibujamos la superficie sobre el dominio del problema. Para hacerlo, puesto que el comando **Plot3D** sólo dibuja sobre parches rectangulares, parametrizamos la superficie en coordenadas polares *x=r Cos(t), y=r Sen(t)*. A continuación, utilizamos el comando **ParametricPlot3D** para dibujarla:

```
superficie = ParametricPlot3D[{r Cos[t], r Sin[t], r^2 Cos[t] Sin[t]},
  {r, 0, 1}, {t, 0, 2 Pi}]
```



El problema corresponde realmente al borde la superficie anterior, dado por el valor *r=1* en la parametrización anterior:

```
linea = ParametricPlot3D[{ Cos[t], Sin[t],  Cos[t] Sin[t], RGBColor[1, 0, 0]},
  {t, 0, 2 Pi}]
```




```
Point[#] & /@ ptos3D
```

$\left\{\text{Point}\left[\left\{-\frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}, \frac{1}{2}\right\}\right], \text{Point}\left[\left\{\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, \frac{1}{2}\right\}\right],\right.$
$\left.\text{Point}\left[\left\{-\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, -\frac{1}{2}\right\}\right], \text{Point}\left[\left\{\frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}, -\frac{1}{2}\right\}\right]\right\}$

Dibujamos los puntos críticos obtenidos anteriormente:

```
puntos = Show[Graphics3D[{PointSize[0.02], RGBColor[0, 0, 1], %}]]
```



Mediante el comando **Show** mezclamos en uno sólo varios dibujos, en este caso, los puntos y la curva del problema:

```
curva = Show[puntos, linea]
```



Ahora consideramos el problema del borde del recinto y el interior. En este caso, el dibujo a considerar es:

```
Show[superficie, curva]
```



y se añade otro punto en el interior del recinto. (¿Cuál?). Discutir el nuevo resultado. Del gráfico se deduce claramente que el nuevo punto es de silla. Esto también se aprecia en la matriz Hessiana. Calcúlese.